Goal Seek and *Solver* in Finance

This document shows how to use *Goal Seek* and *Solver* in some finance settings.

Goal Seek and *Solver* are Excel tools that find solutions to complex problems via iterations (trial and error). While *Goal Seek* allows one variable to change and searches for a specific outcome, *Solver* allows multiple variables to change at once and searches for either a specific outcome or a minimum/maximum. Thus, *Solver* is more flexible and can handle all problems that *Goal Seek* can handle and more, but *Solver* is just a little more complex to use and needs to be installed beforehand as an add-in to Excel.

©ErikLie

PART I: Simple finance applications using *Goal Seek*

1. Estimating YTM for bonds

The value of a bond with semi-annual coupons and a $1,000 face value can be estimated as follows:

$$\text{Bond value} = \frac{\$1{,}000}{(1+r)^n} + \text{Semi-annual coupon payment} \times \left[\frac{1 - {}^{1}\!/\!{(1+r)^n}}{r}\right]$$

where *r* is the semi-annual interest rate and *n* is the number of six-month periods until maturity.

For example, if the semi-annual coupon payments are $50, the semi-annual interest rate is 5.5% (meaning that the annual bond yield is 5.5% × 2 = 11%), and the remaining maturity is 7 years, then the bond value is:

$$\text{Bond value} = \frac{\$1{,}000}{(1+0.055)^{14}} + \$50 \times \left[\frac{1 - {}^{1}\!/\!{(1+0.055)^{14}}}{0.055}\right] = \$952$$

You could also use the function PV(·) in Excel to value the bond, as illustrated in column B of the Excel spreadsheet below.

But what if we already know the bond price to be $952 and instead wish to solve for the interest rate? In that case, we need to solve for *r* in the following expression:

$$\frac{\$1{,}000}{(1+r)^{14}} + \$50 \times \left[\frac{1 - {}^{1}\!/\!{(1+r)^{14}}}{r}\right] = \$952$$

The function INT(·) in Excel can be used solve for the interest rate, as illustrated in column C of the Excel spreadsheet.

	A	B	C	D	E	F	G	H
1		*Find bond value using =PV()*	*Find interest rate using =RATE()*	*Find bond value using formula*	*Find interest rate using formula and Goal Seek*			
2	Face value	$1,000	$1,000	$1,000	$1,000			
3	Semi-annual coupons	$50	$50	$50	$50			
4	Years until maturity	7	7	7	7			
5	Semi-annual interest	5.50%	5.501%	5.50%	5.501%			
6	Bond value	-$952.05	-$952	$952.05	$952			
7								



	A	B	C	D	E
1		*Find bond value using =PV()*	*Find interest rate using =RATE()*	*Find bond value using formula*	*Find interest rate using formula and Goal Seek*
2	Face value	1000	1000	1000	1000
3	Semi-annual coupons	50	50	50	50
4	Years until maturity	7	7	7	7
5	Semi-annual interest	0.055	=RATE(C4*2,C3,C6,C2)	0.055	0.0550055694894129
6	Bond value	=PV(B5,B4*2,B3,B2)	-952	=D2/(1+D5)^(D4*2)+D3*(1-(1/((1+D5)^(D4*2))))/D5	=E2/(1+E5)^(E4*2)+E3*(1-(1/((1+E5)^(E4*2))))/E5

But let us pretend that the specialized functions are not available. We could still estimate the bond value in Excel using the bond valuation formula, as shown in cell D6 of the spreadsheet. Solving for r is a bit trickier. Goal Seek to the rescue. We can request *Goal Seek* to try various interest rates until the bond valuation formula yields the desired value of \$952. After copying column D to column E, select *Data* > *What-if-Analysis* > *Goal Seek*, and in the dialog box, request cell E6 to equal \$952 by changing cell E5. The answer is, once again, 5.5%. That ends our first application of *Goal Seek*.

2. Estimating implied volatility for options

To value the call option on a stock, we can use the Black-Scholes formula:

$$C_0 = S_0 N(d_1) - Xe^{-rT} N(d_2); \quad d_1 = \frac{\ln\left({S_0}/{X}\right) + \left(r + {\sigma^2}/{2}\right)T}{\sigma\sqrt{T}}; \quad d_2 = d_1 - \sigma\sqrt{T}$$

where S_0 is the current stock price, X is the exercise price, r is the risk-free rate, T is the years until maturity, σ is the standard deviation of returns (i.e., volatility), and $N(\cdot)$ is the cumulative area of the standard normal distribution. While the formula looks intimidating, option calculators that employ it are widely available on the internet and it can readily be inserted into a spreadsheet, as illustrated below.

	A	B	C
1	S	\$20.00	Stock price
2	X	\$25.00	Exercise price
3	r	4%	Continuously compounded risk-free rate per year
4	T	1	Time to maturity in years
5	σ	40%	Standard deviation of yearly returns
6			
7	d1	-0.25786	
8	d2	-0.65786	
9			
10	C	\$1.83	Call option value

	A	B
1	S	20
2	X	25
3	r	0.04
4	T	1
5	σ	0.4
6		
7	d1	=(LN(B1/B2)+(B3+B5^2/2)*B4)/(B5*SQRT(B4))
8	d2	=B7-(B5*SQRT(B4))
9		
10	C	=B1*NORMSDIST(B7)-B2*EXP(-B3*B4)*NORMSDIST(B8)

If we already know the option price (also called option premium), it is often useful to estimate the volatility that supports the price. The resulting estimate is called *implied volatility*. The volatility index VIX (also called the fear index) is an example of an implied volatility based on options on the S&P500 index.

Goal Seek can quickly find the implied volatility based on the spreadsheet that calculates the option value. In the *Goal Seek* dialog box, we set cell B10 to the current option premium of, say, \$2 by changing cell B5 that contains the volatility. *Goal Seek* then finds the implied volatility to be 42%.

	A	B	C
1	S	$20.00	Stock price
2	X	$25.00	Exercise price
3	r	4%	Continuously compounded risk-free rate per year
4	T	1	Time to maturity in years
5	σ	40%	Standard deviation of yearly returns
6			
7	d1	-0.25786	
8	d2	-0.65786	
9			
10	C	$1.83	Call option value
11			
12			

Goal Seek
Set cell: B10
To value: 2
By changing cell: B5
OK Cancel

	A	B	C
1	S	$20.00	Stock price
2	X	$25.00	Exercise price
3	r	4%	Continuously compounded risk-free rate per year
4	T	1	Time to maturity in years
5	σ	42%	Standard deviation of yearly returns
6			
7	d1	-0.22349	
8	d2	-0.64516	
9			
10	C	$2.00	Call option value
11			
12			

Goal Seek Status
Goal Seeking with Cell B10 found a solution.
Target value: 2
Current value: $2.00
Step Pause
OK Cancel

3. Estimating IRR

Suppose that a project with a cost of $200K is expected to generate a cash flow of $15K, $25K, $30K, $35K, and $40K over the next five years. The expected growth in cash flow in subsequent years is 3%. The net present value (NPV) of the project can then be estimated as follows:

$$NPV = -\$200{,}000 + \frac{\$15{,}000}{1+r} + \frac{\$25{,}000}{(1+r)^2} + \cdots + \frac{\$40{,}000}{(1+r)^5} + \frac{\$40{,}000 \times 1.03 / r - 0.03}{(1+r)^5}$$

If the discount rate, *r*, is 12%, the NPV equals $159,372. As illustrated below, the NPV(·) function in Excel can assist with this calculation.

	A	B	C	D	E	F	G
1	Year	0	1	2	3	4	5
2	Cash flow	($200,000)	$15,000	$25,000	$30,000	$35,000	$40,000
3	Terminal value						$457,778
4	Sum	($200,000)	$15,000	$25,000	$30,000	$35,000	$497,778
5							
6	Discount rate	12.0%					
7	Growth rate after year 5	3%					
8	NPV	$159,372					

	A	B	C	D	E	F	G
1	Year	0	1	2	3	4	5
2	Cash flow	-200000	15000	25000	30000	35000	40000
3	Terminal value						=(G2*(1+B7))/(B6-B7)
4	Sum	=B2+B3	=C2+C3	=D2+D3	=E2+E3	=F2+F3	=G2+G3
5							
6	Discount rate	0.12					
7	Growth rate after year 5	0.03					
8	NPV	=NPV(B6,C4:G4)+B4					

What if we instead want to estimate the internal rate of return (IRR), i.e., the discount rate that makes the NPV equal to zero? That implies solving for *r* in the following:

$$-\$200{,}000 + \frac{\$15{,}000}{1+r} + \frac{\$25{,}000}{(1+r)^2} + \cdots + \frac{\$40{,}000}{(1+r)^5} + \frac{\$40{,}000 \times 1.03 / r - 0.03}{(1+r)^5} = \$0$$

We can find *r* via trial and error. Had there not been a *terminal value* (the last term that captures the value of the cash flow beyond year 5 based on the growth rate), we could have used the IRR(·) function in Excel to run the trial and error for us. But fortunately, *Goal Seek* still works. In the *Goal Seek* dialog box, we set the cell that estimates the NPV to zero by changing the cell that contains the discount rate. *Goal Seek* then finds the IRR to be 18.3%.

	A	B	C	D	E	F	G
1	Year	0	1	2	3	4	5
2	Cash flow	($200,000)	$15,000	$25,000	$30,000	$35,000	$40,000
3	Terminal value						$457,778
4	Sum	($200,000)	$15,000	$25,000	$30,000	$35,000	$497,778
5							
6	Discount rate	12.0%					
7	Growth rate after year 5	3%					
8	NPV	$159,372					
9							
10							

Goal Seek ? ×
Set cell: B8
To value: 0
By changing cell: B6
OK Cancel

	A	B	C	D	E	F	G
1	Year	0	1	2	3	4	5
2	Cash flow	($200,000)	$15,000	$25,000	$30,000	$35,000	$40,000
3	Terminal value						$269,260
4	Sum	($200,000)	$15,000	$25,000	$30,000	$35,000	$309,260
5							
6	Discount rate	18.3%					
7	Growth rate after year 5	3%					
8	NPV	$0					
9							
10							

Goal Seek Status ? ×
Goal Seeking with Cell B8 found a solution.
Target value: 0
Current value: $0
Step Pause OK Cancel

PART I: Portfolio applications using *Solver*

1. Pet store puzzle

In the first problem, we will go to the pet store to buy a "portfolio" of pets. Admittedly, this is just a silly puzzle, variations of which have been featured on NPR's *Car Talk* and in the *Wall Street Journal*. But it is very useful to illustrate the application of Excel's *Solver*.

At the pet store, we need to buy at least one dog, one cat, and one mouse at prices of $15, $1, and $0.25, respectively. Furthermore, we need to buy a total of 100 animals at a total cost of exactly $100. How many dogs, cats, and mice must we buy?

I have set up the problem in an Excel spreadsheet. The spreadsheet uses the number of dogs, cats, and mice bought and their respective prices to estimate the total number of pets bought and the total price. Because the prices are fixed, we can only vary the number of dogs, cats, and mice. Moreover, because *Goal Seek* can only handle variations in one input variable, we must use *Solver* for this problem.

	A	B	C	D
1	Animals to buy	100		
2	Money to spend	$100		
3				
4		Unit price	Quantity	Total price
5	Dogs	$15	1	$15
6	Cats	$1	1	$1
7	Mice	$0.25	1	$0
8			3	$16

	A	B	C	D
1	Animals to buy	100		
2	Money to spend	100		
3				
4		Unit price	Quantity	Total price
5	Dogs	15	1	=B5*C5
6	Cats	1	1	=B6*C6
7	Mice	0.25	1	=B7*C7
8			=SUM(C5:C7)	=SUM(D5:D7)

To open the *Solver* dialog box, we select *Solver* from the *Data* menu.[1] We have several objectives here, inlcuding purchasing 100 pets and spending $100. But because *Solver* only allows one objective, we must convert other objectives to constraints. That means that there are alternative approaches that work here, depending on what we choose as the objective. I chose to set the objective to spending exactly $100 by changing the number of dogs, cats, and mice we buy. I then include three constraints:

- The number of dogs, cats, and mice we buy must all be
  - at least one, and
  - integers (meaning that we cannot buy, e.g., a half mouse).
- The total pets we buy must equal 100.

Solver then finds the solution of 3 dogs, 41 cats, and 56 mice.

[1] If *Solver* does not appear in the menu, you need to activate it by selecting *File* > *Options* > *Add-ins* > *Solver Add-In* for PC and *Data* > *Analysis Tools* > *Solver Add-In* for Mac.



	A	B	C	D
1	Animals to buy	100		
2	Money to spend	$100		
3				
4		Unit price	Quantity	Total price
5	Dogs	$15	3	$45
6	Cats	$1	41	$41
7	Mice	$0.25	56	$14
8			100	$100

2. Stock portfolio

Let us move on to a more elaborate stock portfolio example. I first downloaded a few years of daily stock price data for US Steel (X), Anheuser-Busch InBev (BUD), Harley-Davidson (HOG), McDonald's (MCD), Coca-Cola (KO), Ford (F), and Walt Disney (DIS) and calculated daily returns, as shown below.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1		Prices							Returns						
2	Date	X	BUD	HOG	MCD	KO	F	DIS	X	BUD	HOG	MCD	KO	F	DIS
3	1/2/2020	10.667	81.180	35.530	188.377	50.419	9.018	148.200							
4	1/3/2020	10.588	80.499	34.754	187.711	50.144	8.817	146.500	-0.007	-0.008	-0.022	-0.004	-0.005	-0.022	-0.011
5	1/6/2020	10.637	81.063	35.013	189.822	50.126	8.769	145.650	0.005	0.007	0.007	0.011	0.000	-0.005	-0.006
6	1/7/2020	10.972	79.390	34.505	190.103	49.741	8.856	145.700	0.032	-0.021	-0.015	0.001	-0.008	0.010	0.000

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1		Prices							Returns						
2	Date	X	BUD	HOG	MCD	KO	F	DIS	X	BUD	HOG	MCD	KO	F	DIS
3	43832	10.66671	81.17964	35.53036	188.3770	50.41933	9.018384	148.1999							
4	43833	10.58785	80.49868	34.75441	187.7109	50.14427	8.817337	146.5	=B4/B3-1	=C4/C3-1	=D4/D3-1	=E4/E3-1	=F4/F3-1	=G4/G3-1	=H4/H3-1
5	43836	10.63714	81.06291	35.01306	189.8218	50.12592	8.769469	145.6499	=B5/B4-1	=C5/C4-1	=D5/D4-1	=E5/E4-1	=F5/F4-1	=G5/G4-1	=H5/H4-1
6	43837	10.97232	79.38970	34.50535	190.1033	49.74083	8.855632	145.6999	=B6/B5-1	=C6/C5-1	=D6/D5-1	=E6/E5-1	=F6/F5-1	=G6/G5-1	=H6/H5-1

Based on the daily stock returns, I calculated daily and annual standard deviation of returns for the seven stocks, as shown below. I also calculated the expected returns for each of the seven stocks using the CAPM, and while these calculations are now shown below, the expected returns are.

	R	S	T	U	V	W	X	Y
1		X	BUD	HOG	MCD	KO	F	DIS
2	Daily std dev	4.3%	2.4%	3.7%	1.8%	1.6%	3.2%	2.4%
3	Ann std dev	68.3%	37.9%	59.1%	27.8%	25.0%	50.4%	37.6%
4	Exp return	12.5%	9.6%	13.6%	9.1%	8.3%	12.1%	10.9%

	R	S	T	U	V	W	X	Y
1		X	BUD	HOG	MCD	KO	F	DIS
2	Daily std dev	=STDEV.S(I:I)	=STDEV.S(J:J)	=STDE	=STD	=STD	=STD	=STD
3	Ann std dev	=S2*SQRT(252)	=T2*SQRT(252)	=U2*S	=V2*	=W2*	=X2*	=Y2*
4	Exp return	0.125	0.096	0.136	0.091	0.08	0.121	0.10

The stock for Coca-Cola has the lowest annual standard deviation at 25%. But I suspect that there is a diversified portfolio that has an even lower standard deviation. Thus, I want to use *Solver* to find the portfolio of the seven stocks that has the lowest standard deviation.

Before unleashing *Solver*, I need a framework for estimating the standard deviation of portfolios based on the seven stocks. To do that, I first created a covariance matrix, as shown below.[2] Based on the covariance matrix and the weights of the portfolio in the seven stocks, I can then estimate the standard

[2] If the covariance matrix gets very large because of many stocks, you can use an add-in tool to create a static covariance matrix by selecting *Data* > *Data Analysis* > *Covariance*. If *Data Analysis* does not appear in the menu, you must first activate it, just like you must do to use *Solver* the first time.

deviation. The spreadsheet shows how this is done for a portfolio that consists of 15% in each of six of the stocks and 10% in Disney.

	Q	R	S	T	U	V	W	X	Y
1			X	BUD	HOG	MCD	KO	F	DIS
2		Daily std dev	4.3%	2.4%	3.7%	1.8%	1.6%	3.2%	2.4%
3		Ann std dev	68.3%	37.9%	59.1%	27.8%	25.0%	50.4%	37.6%
4		Exp return	12.5%	9.6%	13.6%	9.1%	8.3%	12.1%	10.9%
5									
6			15%	15%	15%	15%	15%	15%	10%
7			X	BUD	HOG	MCD	KO	F	DIS
8	15%	X	0.0019	0.0004	0.0007	0.0002	0.0002	0.0006	0.0005
9	15%	BUD	0.0004	0.0006	0.0004	0.0002	0.0002	0.0004	0.0003
10	15%	HOG	0.0007	0.0004	0.0014	0.0003	0.0003	0.0007	0.0005
11	15%	MCD	0.0002	0.0002	0.0003	0.0003	0.0002	0.0003	0.0002
12	15%	KO	0.0002	0.0002	0.0003	0.0002	0.0002	0.0002	0.0002
13	15%	F	0.0006	0.0004	0.0007	0.0003	0.0002	0.0010	0.0004
14	10%	DIS	0.0005	0.0003	0.0005	0.0002	0.0002	0.0004	0.0006
15	100%		0.0001	0.0001	0.0001	0.0000	0.0000	0.0001	0.0000
16									
17		Daily std dev	2.08%						
18		Ann std dev	**33.02%**						
19		Exp return	**10.87%**						
20		Risk-free rate	4.0%						
21		Sharpe ratio	0.208						

	Q	R	S	T	U	V	W	X	Y
1			X	BUD	HOG	MCD	KO	F	DIS
2		Daily std dev	=STDEV.S(I:I)	=STDEV.S(J:J)	=STDE	=STD	=STD	=STD	=STD
3		Ann std dev	=S2*SQRT(252)	=T2*SQRT(252)	=U2*S	=V2*	=W2	=X2*	=Y2*
4		Exp return	0.125	0.096	0.136	0.091	0.08	0.121	0.10
5									
6			=Q8	=Q9	=Q10	=Q11	=Q12	=Q13	=Q14
7			X	BUD	HOG	MCD	KO	F	DIS
8	0.15	X	=COVARIANCE.S($I:$I,I:I)	=COVARIANCE.S($I:$I,J:J)	=COVA	=COV	=COV	=COV	=COV
9	0.15	BUD	=COVARIANCE.S($J:$J,I:I)	=COVARIANCE.S($J:$J,J:J)	=COVA	=COV	=COV	=COV	=COV
10	0.15	HOG	=COVARIANCE.S($K:$K,I:I)	=COVARIANCE.S($K:$K,J:J)	=COVA	=COV	=COV	=COV	=COV
11	0.15	MCD	=COVARIANCE.S($L:$L,I:I)	=COVARIANCE.S($L:$L,J:J)	=COVA	=COV	=COV	=COV	=COV
12	0.15	KO	=COVARIANCE.S($M:$M,I:I)	=COVARIANCE.S($M:$M,J:J)	=COVA	=COV	=COV	=COV	=COV
13	0.15	F	=COVARIANCE.S($N:$N,I:I)	=COVARIANCE.S($N:$N,J:J)	=COVA	=COV	=COV	=COV	=COV
14	0.1	DIS	=COVARIANCE.S($O:$O,I:I)	=COVARIANCE.S($O:$O,J:J)	=COVA	=COV	=COV	=COV	=COV
15	=SUM(Q8:Q14)		=S6*SUMPRODUCT($Q8:$Q14,S8:S14)	=T6*SUMPRODUCT($Q8:$Q14,T8:T14)	=U6*S	=V6*	=W6	=X6*	=Y6*
16									
17		Daily std dev	=SQRT(SUM(S15:Y15))						
18		Ann std dev	**=S17*SQRT(252)**						
19		Exp return	**=SUMPRODUCT(S4:Y4,S6:Y6)**						
20		Risk-free rate	0.04						
21		Sharpe ratio	=(S19-S20)/S18						

Now I am ready to use *Solver*. In *Solver's* dialog box, I minimize the cell containing the portfolio standard deviation by changing the cells containing the portfolio weights, subject to the constraints that the portfolio weights must be positive and add up to 100%. *Solver* found a portfolio of stocks (called the *minimum variance portfolio*) based on Coca-Cola (60%), McDonald's (34%), and Anheuser-Busch (6%)

that has a standard deviation of 23.46% and an expected return of 8.66%. This shows that I was indeed able to create a portfolio with a lower standard deviation than Coca-Cola by itself.



0%	X	0.0019
6%	BUD	0.0004
0%	HOG	0.0007
34%	MCD	0.0002
60%	KO	0.0002
0%	F	0.0006
0%	DIS	0.0005
100%		0.0000
	Daily std dev	1.48%
	Ann std dev	**23.46%**
	Exp return	**8.66%**

Suppose that I am not content with the expected return of 8.66% for the *minimum variance portfolio*. So, let us see if we can find the portfolio with the lowest standard deviation that has an expected return of at least 12%. That entails adding a constraint to *Solver* for the expected return, as shown in the dialog box. The resulting portfolio includes US Steel, Harley-Davidson, McDonald's, Ford, and Disney and has a standard deviation of 39.48%.



9%	X	0.0019
0%	BUD	0.0004
32%	HOG	0.0007
8%	MCD	0.0002
0%	KO	0.0002
21%	F	0.0006
30%	DIS	0.0005
100%		0.0001
	Daily std dev	2.49%
	Ann std dev	**39.48%**
	Exp return	**12.00%**

I repeated this process to find portfolios with minimum expected returns ranging from 9% to 13.5% in increments of 0.5%. The standard deviations and expected returns for the seven stocks and the 11 portfolios are given below. For now, you can ignore rows 20–21.

	A	B	C	D
1			Std dev	Exp ret
2	Stocks	X	68.3%	12.5%
3		BUD	37.9%	9.6%
4		HOG	59.1%	13.6%
5		MCD	27.8%	9.1%
6		KO	25.0%	8.3%
7		F	50.4%	12.1%
8		DIS	37.6%	10.9%
9	Portfolios	1	23.5%	8.7%
10		2	23.8%	9.0%
11		3	25.2%	9.5%
12		4	27.3%	10.0%
13		5	29.7%	10.5%
14		6	32.6%	11.0%
15		7	35.8%	11.5%
16		8	39.4%	12.0%
17		9	43.6%	12.5%
18		10	49.2%	13.0%
19		11	56.8%	13.5%
20	CML	Risk-free	0.0%	4.0%
21		Tangent	27.2%	9.99%

Based on the expected returns and standard deviations of the seven stocks and the 11 portfolios, I made a scatter plot.[3] The orange line based on the 11 portfolios represents the *efficient frontier* (i.e., the optimal portfolios) for the seven stocks.



[3] To make a scatter plot with two separate data sets that should be kept apart, first click on a blank cell and select *Insert* > *Scatter*. A blank box appears. To select the data for the plot, select *Chart Design* > *Select Data* > *Add*. For the first series, highlight cell A2 as the series name, cells C2:C8 as the series X values, and cells D2:D8 as the series Y values. For the seconds series, highlight cell A9 as the series name, cells C9:C19 as the series X values, and cells D9:D19 as the series Y values. The plot then appears. To add lines between points in a dataset, click on one of the points, right-click and select *Format Data Series*, and then choose a line and alter other formating as you wish.

I also wish to add the risk-free rate to the plot and make a capital market line (CML).[4] The capital market line is the line between the risk-free rate and a tangent on the orange line that maximizes the slope. The slope of the line is the Sharpe ratio for the portfolio, defined as the difference between the expected portfolio return and the risk-free rate scaled by the portfolio standard deviation. To find the tangent portfolio, I maximize the Sharpe ratio using *Solver*, as shown in the dialog box. The tangent portfolio has a standard deviation of 27.22% and an expected return of 9.99%.



[4] To make a true capital market line, I should include most of the stocks available in my dataset, and not only the seven stocks in my example.

4%	X	0.0019
7%	BUD	0.0004
8%	HOG	0.0007
31%	MCD	0.0002
23%	KO	0.0002
6%	F	0.0006
22%	DIS	0.0005
100%		0.0000
	Daily std dev	1.71%
	Ann std dev	**27.22%**
	Exp return	**9.99%**
	Risk-free rate	4.0%
	Sharpe ratio	0.220

Then I add the dataset consisting of the risk-free security and the tangent portfolio to my plot and make a dashed line between the two points. The resulting plot is shown below.